Panel Graphs and Reports with SAS

Reporting clinical trial results in SAS using panel graphs, data _null_ steps and complex data wrangling of a long dataset. This report uses mock and camouflaged data and is for illustrative purposes only. It was submitted as coursework at the university.

Author: Saurabh Gupta
Date: 21May18

Table of Contents

Note: Background of the study is on the next page.

Note: This page has been intentionally left blank.

Panel of clinical trial results



Background

The data for this report comes from the clinical trial of a new diabetes treatment.

The panel on the left indicates in changes in means of vital signs of all subjects who completed the visits. Visit 2 is the baseline while Visit 8 is the last recorded visit for most subjects. On an average, there were 257 subjects in Visit 2 and 235-240 for Visit 8.

For subsequent analyses (shift tables and plots), endpoint has been regarded as the latest recorded visit after Visit 2 (i.e. if VSNO > 2 and last.vsno in the SAS code).

For HbA1c, the value for a retest was used for the analysis, if there was a retest for any of the visits.

The Treatments were coded as:

- "Old" if T_ASTRGRP = 36
- "New" if T_ASTRGRP = 901

The SAS code and a frequency table of the number of subjects by each treatment group, lab variable and visit have been provided in the appendix.

Key observations on the means of vistal signs

HbA1c reduced more gently for New Treatment group compared to Old. Overall, decrease was of 1 point from Visit 2 to Visit 7.

Systolic blood pressure reduced with less variability (continuous downward slope) in the New Treatment compared to Old Treatment. For the Old Treatment, it went slightly up in visit number 4. Overall variability was small in the range 133 ± 2 units.

Diastolic blood pressure was more erratic compared to Systolic for the New Treatment. However, the differences appear to be very small at 76 ± 1 unit.

Heart rate was more erratic for the Old Treatment. For the New Treatment it was lower through the visits compared to visits 1 and 2. Overall decrease was of more than 1 unit.

Weight (kg) went up in both treatments. However, the increase is apparently less for the New Treatment (the slope is slightly more gentle). Overall increase is below 2 kgs for both groups.

BMI also showed a similar trend. Overall, the BMI was higher for the new Treatment while the weight was lower. It indicates that on average the subjects were shorter for the New treatment, though the difference isn't large.

Reports using DATA _null_ steps

Following pages show shifts in values of different blood tests for two different diabetes treatments.



Table 5.6
Laboratory Normal Range Shift Table

BASOPHIL (Lab Units:3)		Baseline Value					
		New Treatment			Standard Treatment		
		Low	Normal	High	Low	Normal	High
End of Study	Low	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)
	Normal	0(0.0%)	239(97.6%)	6(2.4%)	0(0.0%)	247(97.6%)	3(1.2%)
	High	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)	3(1.2%)	0(0.0%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- For the New Treatment no subjects experienced this shift



Table 5.6
Laboratory Normal Range Shift Table

BILITOT (Lab Units:95)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	0(0.0%)	2(0.8%)	0(0.0%)	0(0.0%)	2(0.8%)	0(0.0%)
	Normal	3(1.2%)	233(94.3%)	2(0.8%)	4(1.6%)	239(94.8%)	3(1.2%)
	High	0(0.0%)	5(2.0%)	2(0.8%)	0(0.0%)	2(0.8%)	2(0.8%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- For the New Treatment 2% subjects experienced this shift compared to 1% for Old

Shift from Normal range in baseline to Low range at endpoint:

- It was similar at 0.8% for both treatments



Table 5.6
Laboratory Normal Range Shift Table

CHOLEST (Lab Units:66)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	40(16.3%)	31(12.6%)	0(0.0%)	58(22.9%)	31(12.3%)	0(0.0%)
	Normal	17(6.9%)	145(58.9%)	6(2.4%)	19(7.5%)	136(53.8%)	1(0.4%)
	High	0(0.0%)	6(2.4%)	1(0.4%)	0(0.0%)	4(1.6%)	4(1.6%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- For the New Treatment it was slightly higher (2.4%) compared to Old (1.6%)

Shift from Normal range in baseline to Low range at endpoint:

- It was similar at approximately 12.5% for both treatments



Table 5.6
Laboratory Normal Range Shift Table

CREATIN (Lab Units:95)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)	0(0.0%)
	Normal	0(0.0%)	231(93.9%)	4(1.6%)	0(0.0%)	231(91.3%)	5(2.0%)
	High	0(0.0%)	3(1.2%)	8(3.3%)	0(0.0%)	6(2.4%)	11(4.3%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- For the New Treatment it was lower (1.2%) compared to Old (2.4%)



Table 5.6
Laboratory Normal Range Shift Table

HB (Lab Units:66)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	8(3.3%)	11(4.5%)	0(0.0%)	6(2.4%)	9(3.5%)	0(0.0%)
	Normal	4(1.6%)	217(88.6%)	2(0.8%)	4(1.6%)	226(89.0%)	4(1.6%)
	High	0(0.0%)	2(0.8%)	1(0.4%)	0(0.0%)	1(0.4%)	4(1.6%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- For both treatments, it was similar at less than 1%

Shift from Normal range in baseline to Low range at endpoint:

- It was slightly higher for New Treatment at 4.5% compared Old (3.5%)



Table 5.6
Laboratory Normal Range Shift Table

PGLUCF (Lab Units:66)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	0(0.0%)	3(1.2%)	5(2.0%)	0(0.0%)	1(0.4%)	7(2.8%)
	Normal	1(0.4%)	10(4.0%)	42(16.8%)	3(1.2%)	6(2.4%)	33(13.0%)
	High	5(2.0%)	14(5.6%)	170(68.0%)	5(2.0%)	15(5.9%)	183(72.3%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from Normal range in baseline to High range at endpoint:

- Similar with just below 6% for both treatments

Shift from Normal range in baseline to Low range at endpoint:

- Similar for both treatments at around 1%

Shift from *High range* in baseline to Normal range at endpoint:

- It was higher for New Treatment at 17% compared Old (13%)



Table 5.6
Laboratory Normal Range Shift Table

PLATELET (Lab Units:11)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	6(2.5%)	1(0.4%)	0(0.0%)	3(1.2%)	4(1.6%)	0(0.0%)
	Normal	1(0.4%)	224(92.6%)	8(3.3%)	1(0.4%)	237(94.8%)	3(1.2%)
	High	0(0.0%)	1(0.4%)	1(0.4%)	0(0.0%)	0(0.0%)	2(0.8%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from *High range* in baseline to Normal range at endpoint:

- It was slightly higher for New Treatment at 3.3% compared Old (1.2%)



Table 5.6
Laboratory Normal Range Shift Table

RBC (Lab Units:20)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	5(2.0%)	11(4.5%)	0(0.0%)	6(2.4%)	5(2.0%)	0(0.0%)
	Normal	5(2.0%)	220(89.8%)	1(0.4%)	6(2.4%)	230(90.6%)	1(0.4%)
	High	0(0.0%)	2(0.8%)	1(0.4%)	0(0.0%)	2(0.8%)	4(1.6%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from *Normal range* in baseline to Lowrange at endpoint:

- It was higher for New Treatment at 4.5% compared Old (2%)



Table 5.6
Laboratory Normal Range Shift Table

WBC (Lab Units:11)		Baseline Value: New Treatment Low	New Treatment Normal	New Treatment High	Standard Treatment Low	Standard Treatment Normal	Standard Treatment High
End of Study	Low	7(2.9%)	5(2.0%)	0(0.0%)	3(1.2%)	3(1.2%)	0(0.0%)
	Normal	6(2.4%)	216(88.2%)	4(1.6%)	8(3.1%)	226(89.0%)	6(2.4%)
	High	0(0.0%)	5(2.0%)	2(0.8%)	0(0.0%)	7(2.8%)	1(0.4%)

Created by C:\Users\saura\Dropbox\STATS701\Data\A2_sgup072.sas on May 21, 2018 using 9.4

Comments

Shift from *Normal range* in baseline to High range at endpoint:

- It was slightly lower for New Treatment at 2% compared Old (2.8%)

Appendix and Code

Frequency Table for Question 1

STATS 701: sgup072
Distribution of Subjects by Treatment Groups, Lab Variables and Visits.

The FREQ Procedure

Table 1 of VARNAME by VSNO

Controlling for T_ASTRGP=Old

Frequency	VSNO(Visit number)								
VARNAME(Name of variable)	1	2	3	4	5	6	7	8	Total
Body mass index	260	257	253	252	246	244	235	238	1985
Diastolic blood pressure	260	257	256	256	250	249	240	240	2008
HbA1c	260	257	0	0	241	0	239	245	1242
Heart rate	261	259	255	256	250	248	240	241	2010
Systolic blood pressure	260	257	256	256	250	249	240	240	2008
Weight (kg)	260	259	255	254	248	246	237	239	1998
Total	1561	1546	1275	1274	1485	1236	1431	1443	11251

Table 2 of VARNAME by VSNO

Controlling for T_ASTRGP=New

Frequency	VSNO(Visit number)								
VARNAME(Name of variable)	1	2	3	4	5	6	7	8	Total
Body mass index	260	258	254	245	238	235	230	235	1955
Diastolic blood pressure	259	256	253	244	237	236	231	234	1950
HbA1c	260	256	0	0	234	0	227	239	1216
Heart rate	259	257	255	246	237	236	232	235	1957
Systolic blood pressure	259	256	253	244	237	236	231	234	1950
Weight (kg)	260	258	254	246	238	235	230	235	1956
Total	1557	1541	1269	1225	1421	1178	1381	1412	10984

SAS code - Setup and Data Exploration

```
*************************************************************;
* program: panel_data_null.sas
      ;
* path: H:\panelreports                                      ;
*
                   ;
* author: Saurabh Gupta  UPI: sgup072                        ;
* Date: 21st May 2018                                  ;
*
                   ;
*************************************************************;


************************************************************ ;
            /* This is the start up code */

*Set up title;
TITLE1 "STATS 701: &sysuserid" ;

*Specify options, define some macros and macro variables    ;
OPTIONS NOFMTERR NONUMBER NODATE FORMCHAR = '|_'
mcompilenote=all merror mexecnote mlogic mprint ;

* Format the page for better printing;
 OPTIONS linesize=100 pagesize=55 ;

/*%LET drive = H: ;           *     select drive according to which
computer being used     ;*/

%LET drive = C: ;

* Change topdir before running on Lab PCs;
/*%LET topdir = &drive\STATS701 ;*/

%LET topdir = &drive\Users\saura\Dropbox\STATS701 ;

* C:\Users\saura\Dropbox\STATS701\SASscripts ;

%LET source = &topdir\SASscripts ;

*include handy macros;
%INCLUDE "&source\Macros701.txt" ;

*Specify SAS library for current directory;

*     set path;
%LET path = &topdir\Data ;

* set libname for current director;

LIBNAME A2 "&path\A2" ;

*     puts user-defined macro variable names in the SASLOG  ;
%put _USER_ ;

* for visualisation lectures;
libname week10 "&topdir\Lectures Visualization";
```

```
/*goptions reset=all;*/
/**/
/*ods html style=minimal;*/

************************************************************ ;
                          /* Explore and Create Datasets */
************************************************************ ;

* a1c has  2842 observations and 7 variables ;
DATA a1c;
      SET A2.a1cdata;
RUN;

%nprint(a1c, 10)

* lab has 87671 observations and 17 variables ;

DATA lab;
      SET A2.lab_subset;
RUN;

%nprint(lab, 10)
%fcontent(lab)

* The data set WORK.TREAT has 521 observations and 6 variables. ;

DATA treat;
      SET A2.treatment;
RUN;

%nprint(treat, 10)

* The data set WORK.VIT has 26445 observations and 9 variables. ;

DATA vit;
      SET A2. vitsigns;
RUN;

%nprint(vit, 20)

PROC PRINT data = vit (obs = 15);
VAR VARCODE VARNAME;
RUN;



* Create format for treatment groups and panel ;

PROC FORMAT ;
VALUE treatf
      36 = "Old"
      901 = "New" ;

VALUE $ varf
      VS_BPS = 'Systolic blood pressure'
      VTWTKG = 'Weight (kg)'
      VS_HR = 'Heart rate'
      VS_BPD = 'Diastolic blood pressure'
      VS_BMI = 'Body mass index'
      a1c    = 'Hba1c' ;
```

```
VALUE panelf
      1 = 'Systolic blood pressure'
      2 = 'Weight (kg)'
      3 = 'Heart rate'
      4 = 'Diastolic blood pressure'
      5 = 'Body mass index'
      6 = 'Hba1c' ;
RUN;

%fcontent(vit)

%fcontent(a1c)
```

SAS code - Panel graphs

```
********************************************************** ;
                        /* Panel graphs */
********************************************************** ;

* Panel plot of var - SBP and DBP, Weight and BMI, Heart Rate and a1c ;
* Line graphs over visits and compare treatment groups;

* Need Unique PatID, Treatment Group;
* need to remove those withy no vsno => 2;

/*Merge tables treat, a1c, vit*/
/*Keep only required variables*/
* create variable panel to arrange the plots;

PROC SQL;
      create table t as
            select t_astrgp format = treatf. , suinvid* 1000 + subno as
subjid, t_acdays
            from treat
            where T_ASTRGP in (36, 901)
      ;
      create table v as
            SELECT suinvid* 1000 + subno as subjid,
            VARNAME, VARCODE, VSNO, VTDT, VALUE, BASELINE, CHANGE
    from vit
            WHERE VARCODE in ('VS_BPD', 'VTWTKG',
                  'VS_BPS', 'VS_BMI', 'VS_HR');
      ;
      CREATE TABLE treatvits as
            select coalesce(t.subjid, v.subjid) as subjid, T_ASTRGP ,
                  VARNAME, VARCODE, VSNO, VTDT, VALUE,
      CASE scan(VARCODE, 1, '')
            WHEN 'VS_BPS' THEN 1
            WHEN 'VTWTKG' THEN 2
            WHEN 'VS_HR'  THEN 3
            WHEN 'VS_BPD' THEN 4
            WHEN 'VS_BMI' THEN 5
      END as Panel format = panelf.

      from t, v
            where t.subjid = v.subjid
            order by t_astrgp, subjid, vsno, VTDT
      ;

      create table a as
            select suinvid* 1000 + subno as subjid,
            VSNO, RETEST, LBSMDT as VTDT, LBVA as Value, LBDMDT,
            put('LBVA', $8. ) as VARCODE,
            put('HbA1c', $60.) as VARNAME,
            6 as Panel format = panelf.
      from a1c
      WHERE VSNO in (1, 2, 5, 7, 8)
      ;
      CREATE TABLE treata1c as
            select coalesce(t.subjid, a.subjid) as subjid, T_ASTRGP ,
                  VARNAME, VARCODE, VSNO, VTDT, VALUE, Panel, LBDMDT,
RETEST
```

```
            from t, a
            where t.subjid = a.subjid
            order by t_astrgp, subjid, vsno, VTDT, RETEST
      ;
quit;

%nprint(treatvits, 10)

%nprint(treata1c, 30)

%fcontent(treata1c)

*****************************************;

* subset a1c ;

%sortit(treata1c, subjid vsno vtdt RETEST  );

* check retests and neighbours ;

PROC PRINT DATA = treata1c (obs = 20);
      WHERE subjid in (68001, 68002, 71002, 71003, 83007, 83008)  ;
      RUN;


* subset a1c to keep only the last visit date ;
* There were 2506 observations read from the data set WORK.TREATA1C ;
* The data set WORK.A1C_RETEST has 2458 observations and 8 variables ;

DATA a1c_retest;
      SET treata1c;
      BY subjid vsno vtdt RETEST;
      IF value ^= . ;
      IF last.vsno;
      DROP LBDMDT RETEST ;
run;

%nprint(a1c_retest, 10)

* subset treatvits ;

%sortit(treatvits, subjid varcode vsno vtdt);

* There were 19980 observations read from the data set WORK.TREATVITS.;
* data set WORK.SUBVITS has 19777 observations and 8 variables. ;

DATA subvits;
      SET treatvits;
      BY subjid varcode vsno vtdt;
      IF value ^= . ;
      IF last.vsno;
run;


%nprint(subvits, 10)

%fcontent(subvits)

%fcontent(a1c_retest)

* merge a1c and the rest ;
```

```
DATA allpanel;
      SET subvits a1c_retest;
RUN;


%sortit(allpanel, subjid varcode vsno vtdt)

%nprint(allpanel, 10)
%fcontent(allpanel)

* The data set WORK.ALLPANEL has 22235 observations and 8 variables. ;
* later need to only keep with atleast 1 vsno > 2;


PROC PRINT data= allpanel (obs = 10);
      WHERE VARCODE = 'LBVA';
RUN;


DATA subpanel;
      SET allpanel;
RUN;

%sortit(subpanel, subjid varcode descending vsno descending vtdt)


DATA subpanel;
      SET subpanel;
      BY subjid varcode descending vsno descending vtdt;
      IF first.varcode THEN lastvisit = vsno;
            RETAIN lastvisit;
RUN;

%nprint(subpanel, 30)

DATA subpanel;
      SET subpanel;
      IF lastvisit > 2;
RUN;


* create macro for sgplot ;
%macro pplot(varc, varn, lab);

TITLE1 &varn; TITLE2; footnote; footnote2; footnote3;

proc SGPLOT data= subpanel (WHERE = (varcode = &varc));
      xaxis type=discrete;
      yaxis label = &lab ;
      VLINE VSNO / response = VALUE stat = mean GROUP= T_ASTRGP
GROUPDISPLAY= CLUSTER MARKERS ;
run;
quit;
TITLE1;
%mend pplot;

%put _user_ ;

****************************************;
* Grid Layout for the plots;
```

```
* run macro in each region to generate the plot;


OPTIONS PAPERSIZE = A4 ORIENTATION = LANDSCAPE ;
ods pdf file= "&path\A2\PanelSub_Final20May.pdf" startpage=off style =
journal;
ods listing close;
ods html close;
ods pdf startpage=now;

TITLE1 HEIGHT=14pt      'Clinical trial results of new Diabetes treatment';
TITLE2 HEIGHT=12pt 'Change in Vital Signs and HbA1c by Visit Number';
TITLE3 HEIGHT=12pt 'Comparisons by Treatment (New vs Old)';

footnote justify=left " Line graphs and markers have been staggered
(shifted horizontally) for easy visibility."; * leave a line space ;
footnote2 justify=left "Report by Saurabh Gupta on
%sysfunc(today(),worddate12.) using SAS Version: &sysver";
footnote3 justify=left "Program directory and Name
&topdir\A2_sgup072_20May18.sas ";

ods layout gridded y=20pct
                  columns = 3 column_widths = (40pct 40pct 40pct)
                  column_gutter= 2pct
                  rows = 2 row_heights=(30pct 30pct)
                  row_gutter = 2pct
                  advance = proc
                  ;

ods region column= 1 ;

%pplot('VS_BPS', 'Systolic blood pressure', 'Mean SBP')

%pplot('VTWTKG', 'Weight (kg)', 'Mean Weight')

%pplot('VS_HR', 'Heart rate', 'Mean Heart Rate')

%pplot('VS_BPD', 'Diastolic blood pressure', 'Mean DBP' )

%pplot('VS_BMI', 'Body mass index', 'Mean BMI')

%pplot('LBVA', 'HbA1c', 'Mean HbA1c')

ods layout end;

ods pdf close;
ods listing;
ODS HTML;

TITLE1; TITLE2;
footnote; footnote2; footnote3;


*********************************************************;
```

SAS code - Shift Tables using DATA _null_ steps

```
************************************************;
/* Shift Tables */
************************************************;


* Need Unique PatID, Treatment Group;
*Keep only required variables;

PROC SQL;
      CREATE TABLE t as
            select t_astrgp format = treatf. , suinvid* 1000 + subno as
subjid, t_acdays
            from treat
            where T_ASTRGP in (36, 901)
      ;
      CREATE TABLE lbase as
            SELECT suinvid* 1000 + subno as subjid,
            LBTSCD, LOBDSI, UPBDSI, VSNO, LBSMDT, LBSIVA, LBSIUN, AGE,
AGEMN, AGEMX
    from lab
            WHERE VSNO = 2 and
            LBTSCD in ('BASOPHIL', 'BILITOT', 'CHOLEST', 'CREATIN',
            'HB', 'PGLUCF', 'PLATELET', 'RBC', 'WBC')
            ;
      ;
      CREATE TABLE lend as
            SELECT suinvid* 1000 + subno as subjid,
            LBTSCD, LOBDSI, UPBDSI, VSNO, LBSMDT, LBSIVA, LBSIUN, AGE,
AGEMN, AGEMX
    from lab
            WHERE VSNO > 2 and
            LBTSCD in ('BASOPHIL', 'BILITOT', 'CHOLEST', 'CREATIN',
            'HB', 'PGLUCF', 'PLATELET', 'RBC', 'WBC')
            ;

      CREATE TABLE trbase as
            select coalesce(t.subjid, lbase.subjid) as subjid, *
            from t, lbase
            where t.subjid = lbase.subjid
            order by subjid, LBTSCD, VSNO, LBSMDT
      ;

      create table trend as
            select coalesce(t.subjid, lend.subjid) as subjid, *
            from t, lend
            where t.subjid = lend.subjid
            order by subjid, LBTSCD, VSNO, LBSMDT
      ;

quit;

%nprint(trbase, 10)

%nprint(trend, 30)

%fcontent(trend)
```

```
* subset datasets - keep last visit only;

%sortit(trbase, subjid LBTSCD VSNO LBSMDT);

DATA subtrbase;
      SET trbase;
      WHERE LBSIVA ^= . ;
      BY subjid LBTSCD VSNO LBSMDT;
      IF last.LBTSCD;
run;


%sortit(trend, subjid LBTSCD VSNO LBSMDT);

DATA subtrend;
      SET trend;
      WHERE LBSIVA ^= . ;
      BY subjid LBTSCD VSNO LBSMDT;
      IF last.LBTSCD;
run;


* create formats ;


PROC FORMAT;
      VALUE $ paramf

      BASOPHIL  = 'Basophils'
      BILITOT = 'Total Bilirubin'
      CHOLEST = 'Cholesterol'
      CREATIN = 'Creatinine'
      HB = 'Haemoglobin'
      PGLUCF = 'Fasting Plasma Glucose'
      PLATELET = 'Platelets'
      RBC = 'Red Blood cell count'
      WBC = 'White blood cells'
      ;

      VALUE treatf
      36 = "Old"
      901 = "New" ;

      VALUE levelf
      1 = "Low"
      2 = "Normal"
      3 = "High" ;

RUN;

* create indicators for hi med low;

%nprint(subtrend, 10)

DATA endpoint;
      SET subtrend;
      endlev = 2;
      IF LBSIVA > UPBDSI THEN endlev = 3 ;
      IF LBSIVA < LOBDSI THEN endlev = 1 ;
      FORMAT endlev levelf. ;
```

```
RUN;


DATA baseline;
      SET subtrbase;
      baselev = 2;
      IF LBSIVA > UPBDSI THEN baselev = 3 ;
      IF LBSIVA < LOBDSI THEN baselev = 1 ;
      FORMAT baselev levelf. ;
RUN;

%freqtest(endpoint, endlev * T_ASTRGP)

%freqtest(baseline, baselev * T_ASTRGP)

%fcontent(endpoint)

%fcontnt(baseline)


* Each  lab  test  should have an original result category and baseline
result category ;

* Merge baseline and endpoint ;
* keep only with vsno > 2 so use inner join;

PROC SQL;
CREATE TABLE endbase as
            select coalesce(e.subjid, b.subjid) as subjid,
                  coalesce(e.LBTSCD, b.LBTSCD) as LBTSCD ,
                  coalesce(e.LBSIUN, b.LBSIUN) as     LBSIUN,
                  coalesce(e.T_ASTRGP, b.T_ASTRGP) as T_ASTRGP ,

                  b.LBSIVA as baseline, e.LBSIVA as endpoint ,
                  baselev, endlev,
                  b.VSNO as basevsno, e.vsno as endvsno

            from endpoint e, baseline b
            where e.subjid = b.subjid and e.LBTSCD = b.LBTSCD
            order by subjid, LBTSCD
      ;
QUIT;

%nprint(endbase, 40)

DATA A2.endbase;
      SET endbase;
RUN;


* check lab names and lab units for PROC Report;
* each variable has only 1 unit - obviously;
* but same unit may be used for > 1 variables;

%freqtest(endbase, LBTSCD * LBSIUN)

* check freq of treatment by levels;
* output is satisfactory;
*  ;
noprint ;
%nprint(endbase, 20)
```

```
%fcontent(endbase)

* Run macro created (in the next section) ************************;

%reportit(BASOPHIL)

%reportit(BILITOT)

%reportit(CHOLEST)

%reportit(CREATIN)

%reportit(HB)

%reportit(PGLUCF)

%reportit(PLATELET)

%reportit(RBC)

%reportit(WBC)


* need a macro to run it on one variable at a time and output its report;

%macro reportit(varn);

proc summary data= endbase (where = ( LBTSCD = "&varn")) nway
completetypes;
   class LBTSCD LBSIUN T_ASTRGP endlev baselev / preloadfmt order=data
missing;
   output out=counts;
   run;


PROC SQL;
      CREATE TABLE f36 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct,

      put(put(_FREQ_, 5.) || compress("(" || put(calculated pct , 4.1 )||
"%)"), $14.-R)
      as props

      FROM counts
      WHERE T_ASTRGP = 36;

      CREATE TABLE f901 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct,

      put(put(_FREQ_, 5.) || compress("("||put(calculated pct , 4.1 )||
"%)"), $14.-R )
      as props

      FROM counts
      WHERE T_ASTRGP = 901;
```

```
      ;
quit;

DATA FINAL;
      SET f36 f901;
RUN;

%sortit(final, endlev descending T_ASTRGP baselev)

DATA final;
      SET final;
      col = put(put(T_ASTRGP, treatf.) || " " || put(baselev, levelf.),
$32.) ;
RUN;

* check data;
%allprint(final);

PROC TRANSPOSE data = final (drop = _TYPE_ _FREQ_ N pct )
            out = report (drop = _name_)
            ;
       ID col;
      BY endlev ;
      VAR props;
RUN;

DATA REPORT;
      FORMAT Col1 $15. endlev levelf.
            New_Low $15.-r New_Normal $15.-r New_High $15.-r
            Old_Low $15.-r Old_Normal $15.-r Old_High $15.-r ;
      SET REPORT;
      IF _n_ = 1 THEN Col1 = 'End of Study';
      IF _n_ > 1 THEN Col1 = '';
      RUN;

%allprint(report )

DATA _NULL_;
      SET final;
      IF _n_ = 1 THEN DO;
            call symput('labname', compress(put(LBTSCD, $16.))) ;
            call symput( 'labunit', put( "(Lab
Units:"||compress(put(LBSIUN, 4.)||")") , $16.));
      END;
      RUN;

* Format the page for better printing;
GOPTIONS reset=all;
TITLE1; TITLE2;
FOOTNOTE1; FOOTNOTE2;

 OPTIONS  PAPERSIZE = A4 ORIENTATION = LANDSCAPE
             LEFTMARGIN= 3cm;

 ODS PDF File= "&path\A2\&labname.pdf" style = journal
            ;

DATA _NULL_ ;
      SET report end = eof;
      FILE PRINT ;
```

```
      IF _n_ = 1 THEN DO;

            PUT @45  'Table 5.6'
                ;
            PUT @33  'Laboratory Normal Range Shift Table'
                ;

            PUT @1 14* '_' @16 11* '_'
                @28  11* '_' @40  11* '_' @52  11* '_'
                @64 11* '_' @76 11* '_' @88 11* '_'
                ;
            PUT @55 'Baseline Value';

            PUT @1 "&labname" @16 11* '_'
                @28  11* '_' @40  11* '_' @52  11* '_'
                @64 11* '_' @76 11* '_' @88 11* '_'
                ;

            PUT @1 "&labunit" @32 'New Treatment' @60 'Standard Treatment'
;

            PUT @28  11* '_' @40  11* '_' @52  11* '_'
                @64 11* '_' @76 11* '_' @88 11* '_'
                ;

            PUT @28  ' Low ' @40  ' Normal ' @52  ' High '
                @64 ' Low '  @76 ' Normal '  @88 ' High '
                ;

            PUT @1 14* '_' @16 11* '_'
                @28  11* '_' @40  11* '_' @52  11* '_'
                @64 11* '_' @76 11* '_' @88 11* '_'
                ;
      END;

            IF endlev ne lag1(endlev) then put @1 Col1 @16 endlev
                @28  New_Low @40  New_Normal @52  New_High
                @64 Old_Low @76 Old_Normal @88 Old_High ;
            IF eof THEN DO;
                PUT @1 100 * '_' ;
                PUT @1 "Created by &path\A2_sgup072.sas on
%sysfunc(today(),worddate12.) using &sysver";
                PUT @1 100 * '_' ;
            END;
RUN;
ODS PDF CLOSE;

%mend reportit;


* macro ends here **************;



* create and test macro ***************************;


proc summary data= endbase (where = ( LBTSCD = 'BASOPHIL')) nway
completetypes;
   class LBTSCD LBSIUN T_ASTRGP endlev baselev / preloadfmt order=data
missing;
```

```
   output out=counts;
   run;
proc print;
   run;

   %fcontent(counts)

PROC SQL;
      CREATE TABLE f36 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct,

      put(put(_FREQ_, 5.) || compress("(" || put(calculated pct , 4.1 )||
"%)"), $14.-R)
      as props

      FROM counts
      WHERE T_ASTRGP = 36;

      CREATE TABLE f901 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct,

      put(put(_FREQ_, 5.) || compress("("||put(calculated pct , 4.1 )||
"%)"), $14.-R )
      as props

      FROM counts
      WHERE T_ASTRGP = 901;
      ;
quit;

DATA FINAL;
      SET f36 f901;
RUN;


%sortit(final, endlev descending T_ASTRGP baselev)

DATA final;
      SET final;
      col = put(put(T_ASTRGP, treatf.) || " " || put(baselev, levelf.),
$32.) ;
RUN;

%allprint(final);

%fcontent(final);


PROC TRANSPOSE data = final (drop = _TYPE_ _FREQ_ N pct ) out = report
(drop = _name_)
       ;
       ID col;
      BY endlev ;
      VAR props;
RUN;

DATA REPORT;
```

```
      FORMAT Col1 $15. endlev levelf.
            New_Low $15.-r New_Normal $15.-r New_High $15.-r
            Old_Low $15.-r Old_Normal $15.-r Old_High $15.-r ;
      SET REPORT;
      IF _n_ = 1 THEN Col1 = 'End of Study';
      IF _n_ > 1 THEN Col1 = '';
      RUN;

      %allprint(report )

DATA _NULL_;
      SET final;
      IF _n_ = 1 THEN DO;
            call symput('labname', compress(put(LBTSCD, $16.))) ;
            call symput( 'labunit', put( "(Lab
Units:"||compress(put(LBSIUN, 4.)||")") , $16.));
      END;
      RUN;

%put _user_ ;

* Format the page for better printing;
ORIENTATION = Landscape;

GOPTIONS reset=all;
TITLE1;
TITLE2;

 OPTIONS  PAPERSIZE = A4 ORIENTATION = LANDSCAPE
            LEFTMARGIN= 3cm;

 ODS PDF File= "&labname.pdf" style = journal
            ;

DATA _NULL_ ;
      SET report end = eof;
      FILE PRINT ;

      IF _n_ = 1 THEN DO;

            PUT @45  'Table 5.6'
                  ;
            PUT @33  'Laboratory Normal Range Shift Table'
                  ;

            PUT @1 14* '_' @16 11* '_'
                  @28  11* '_' @40  11* '_' @52  11* '_'
                  @64 11* '_' @76 11* '_' @88 11* '_'
                  ;
            PUT @55 'Baseline Value';

            PUT @1 "&labname" @16 11* '_'
                  @28  11* '_' @40  11* '_' @52  11* '_'
                  @64 11* '_' @76 11* '_' @88 11* '_'
                  ;

            PUT @1 "&labunit" @32 'New Treatment' @60 'Standard Treatment'
;

            PUT @28  11* '_' @40  11* '_' @52  11* '_'
                  @64 11* '_' @76 11* '_' @88 11* '_'
```

```
                  ;

            PUT @28  ' Low ' @40  ' Normal ' @52  ' High '
                  @64 ' Low '  @76 ' Normal '  @88 ' High '
                  ;

            PUT @1 14* '_' @16 11* '_'
                  @28  11* '_' @40  11* '_' @52  11* '_'
                  @64 11* '_' @76 11* '_' @88 11* '_'
                  ;
      END;

            IF endlev ne lag1(endlev) then put @1 Col1 @16 endlev
                  @28  New_Low @40  New_Normal @52  New_High
                  @64 Old_Low @76 Old_Normal @88 Old_High ;
            IF eof THEN DO;
                  PUT @1 100 * '_' ;
                  PUT @1 "Created by &path\A2_sgup072.sas on
%sysfunc(today(),worddate12.) using &sysver";
                  PUT @1 100 * '_' ;
            END;
RUN;
ODS PDF CLOSE;

******** end of macro test ******************;

************ End of shift tables ******************;
```

SAS code - shift plots

```
/*****************************************************/
* shift plots ;
/*****************************************************/

* create formats;

PROC FORMAT ;
VALUE treatf
      36 = "Old"
      901 = "New" ;
RUN;

* read dataset created for question 2;

DATA endbase ;
      SET A2.endbase;
      FORMAT T_ASTRGP treatf. endpoint 5.1 baseline 5.1 ;
      LABEL T_ASTRGP = 'Treatment Group';
RUN;

* check dataset;
%fcontent(endbase)

%nprint(endbase, 10)

* Run macro to create frequency plots;
* code for macro given in the next section;

%plotfreq(BASOPHIL)


%plotfreq(BILITOT)


%plotfreq(CHOLEST)

%plotfreq(CREATIN)

%plotfreq(HB)

%plotfreq(PGLUCF)

%plotfreq(PLATELET)

%plotfreq(RBC)

%plotfreq(WBC)



* Create macro for plots;

%macro plotfreq(varn);

* get a summary so that all levels are included;
```

```
proc summary data= endbase (where = ( LBTSCD = "&varn")) nway
completetypes;
   class LBTSCD T_ASTRGP endlev baselev / preloadfmt order=data missing;
   output out=counts;
   run;

 * calculate table percent for each treatment ;
PROC SQL;
      CREATE TABLE p36 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct

      FROM counts
      WHERE T_ASTRGP = 36;

      CREATE TABLE p901 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct

      FROM counts
      WHERE T_ASTRGP = 901;
      ;
quit;

* and concatenate the two datasets ;

DATA plottab;
      SET p36 p901;
RUN;

* sort for us in sgpanel;

%sortit(plottab, T_ASTRGP baselev endlev )

* only create plots where baseline is not same as endpoint;
PROC SGPANEL data= plottab (where = ( endlev ne baselev ));
      * treatment group side by side ;
      PANELBY T_ASTRGP / sort = DESCENDING;
      * labels for the variables to be shown on the panel ;
      LABEL pct= 'Percent of Subjects';
      LABEL T_ASTRGP = 'Treatment';
      LABEL endlev = 'Level of Lab Values at Endpoint';
      LABEL baselev = 'Level of Lab Values at Baseline';

      * plot end levels by category of baseline level;
      vbarparm CATEGORY= baselev response = pct /  group = endlev
                        GROUPDISPLAY=CLUSTER GROUPORDER = ASCENDING ;

      * Title and subtitle for the panel;
      TITLE2 "Shifts in Lab Values of &varn";
      TITLE3 'By Treatment Group';
      * add foot notes for easy reference;
      footnote justify=left  "Plot includes only levels where baseline
differs from endpoint."; * leave a line space ;
      footnote2 justify=left "Report by Saurabh Gupta on
%sysfunc(today(),worddate12.)";
      footnote3 justify=left "Program: &topdir\A2_sgup072_20May.sas run on
SAS Version: &sysver ";
run;
```

```
quit;


%mend plotfreq;



************************************************************;

* Test Macro ;

* proc summary ;

proc summary data= endbase (where = ( LBTSCD = "BASOPHIL")) nway
completetypes;
   class LBTSCD LBSIUN T_ASTRGP endlev baselev / preloadfmt order=data
missing;
   output out=counts;
   run;

   %allprint(counts)


PROC SQL;
      CREATE TABLE p36 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct

      FROM counts
      WHERE T_ASTRGP = 36;

      CREATE TABLE p901 AS
      SELECT *,
      sum(_FREQ_) as N,
      100 * _FREQ_ / calculated N as pct

      FROM counts
      WHERE T_ASTRGP = 901;
      ;
quit;

DATA plottab;
      SET p36 p901;
RUN;

%sortit(plottab, T_ASTRGP baselev endlev )

%allprint(plottab)

PROC SGPANEL data= plottab (where = ( endlev ne baselev ));
      * class variables to be used for the plots ;
      * 3 columns and 2 rows ;
      PANELBY T_ASTRGP / sort = descending ;
      * labels for the variables to be shown on the panel ;
      LABEL pct= 'Percent of Subjects';
      LABEL T_ASTRGP = 'Treatment';
      LABEL endlev = 'Level of Lab Values at Endpoint';
      LABEL baselev = 'Level of Lab Values at Baseline';

      vbarparm CATEGORY= baselev response = pct /  group = endlev
```

```
                        GROUPDISPLAY=CLUSTER GROUPORDER = ASCENDING ;

      * Title and subtitle for the panel;
      TITLE2 'Shifts in Abnormality of Lab Values';
      TITLE3 'By Treatment Group';
      * add foot notes for easy reference;
      footnote justify=left  "Plot includes only levels where baseline
differs from endpoint."; * leave a line space ;
      footnote2 justify=left "Report by Saurabh Gupta on
%sysfunc(today(),worddate12.)";
      footnote3 justify=left "Program: &topdir\A2_sgup072_20May.sas run on
SAS Version: &sysver ";
run;
quit;


%allprint(plottab)



************************************************************;
```